
☆ 1
 General-Purpose computing on Graphics Processing Units.

マルチコア時代のプログラミング

　コンピュータの性能向上は長らく，クロックの高

速化やパイプラインの改善によって達成されてきた．

しかし，シングルプロセッサの性能向上は頭打ちと

なり，古きよき時代は終わりを告げた．最新のプロ

セッサは，複数の実行コアを備えたマルチコア構成

をとるものが普通である．また，グラフィクス処理

用チップの多数の演算ユニットを汎用計算処理に用

いる GPGPU
☆ 1も身近になってきている．近い将

来には，これらを複数搭載したマルチプロセッサマ

シンが高速ネットワークで相互接続された大規模並

列分散環境が一般的なものになっているだろう．

　しかし，そのような環境では実行コアの総数は数

千～数十万個にもなるため，それらを活用できるソ

フトウェアを開発することは容易ではない．そのた

めには，ハードウェア環境の変革に合わせ，プログ

ラミング環境も「マルチコア時代」へと向けた変革を

遂げることが必要となる．具体的な要件としては，

以下のものが挙げられる．

　1.大量の非同期タスクを軽量に生成・消滅でき

ること．また，デッドロック等に悩まされず

にタスク間の同期が行えること．

　2. 共有メモリ環境と分散メモリ環境を単一のプ

ログラムで扱えること．また，異なるアーキ

テクチャの実行環境をサポートできること．

　3. 複数マシン間の複雑な通信処理を記述しなく

てよいこと．しかし，並列処理や分散処理を

ある程度明示的に制御できること．

　4. 開発をサポートするツール群が用意されてい

ること．

これらの要件を満たすべく IBM Researchが開発中

最新のプロセッサでは複数の実行コアを備えた「マルチコア」が一般的になっている．しかし，多数のマルチコア

CPU や GPGPU などが相互接続された並列分散環境を活用するためのプログラミングは容易ではない．このような並

列分散プログラミングの生産性を向上させることを目標とした，新しいプログラミング言語が X10（エックステン）

である．

　X10 はいわゆる PGAS（Partitioned Global Address Space）を採用している言語族の一員であり，複数の「プレー

ス」にまたがったデータ構造を扱える．さらに，軽量な並列タスクを生成するための「async」や分散処理のための

「at」，同期のための「atomic」や「clocked」などの構文を提供している．本稿では，X10 の設計思想やプログラ

ミングモデルに加え，これらの特徴的な構文についても使用例を交えつつ解説する．

解 説

☆ 2
 Defense Advanced Research Projects Agency.

☆ 3
 High Productivity Computing Systems.

☆ 4
 Productive Easy-to-use Reliable Computer Systems.

☆ 5
 X10という名前は，「（並列アプリケーションの）生産性を 10倍向上させる」という目標からきている．

の新しいプログラミング言語が，X10（エックス

テン）である．本稿では，X10の設計思想やプログ

ラミングモデル，データ型や文法について，X10 

2.1.2の仕様をベースに解説を行う．掲載したプロ

グラムはすべて単体でコンパイル・実行が可能な形

となっているので，X10アプリケーション開発時

のサンプルとして活用していただければ幸いである．

X10 の概要

　X10は，IBM Researchが開発している新しい並

列分散プログラミング言語で，米 DARPA
☆ 2の

HPCS
☆ 3プログラムに基づく IBMの PERCS

☆ 4

プロジェクトの一部である．PERCSプロジェク

トは，先進的なチップ技術とアーキテクチャ，OS，

プログラミング言語などを統合し，ハードウェアと

ソフトウェアを総合的にデザインすることで，並列

アプリケーションの生産性を向上させることを目標

としている☆ 5．そのために，X10は新しいプログ

ラミングモデルと開発ツール群を提供する．

　X10の開発は 2004年に開始され，オープンソー

スプロジェクトとして行われている．ライセンス形

態は Eclipse Public Licenseである．成果は http://

x10-lang.org/
1）からアクセス可能で，ここから最新

の実行環境とそのソースコードに加え，言語仕様 2）

やチュートリアル等も取得できる．メーリングリス

トなどによるサポートも行われている．

 ❏ PGASプログラミングモデル
　さて，多数の実行コアを備えた並列分散環境にお

けるプログラミングでは，その並列性やメモリ構成

をプログラマにどのように見せるかの選択が重要に

なる（図 -1上）．並列性を隠蔽し処理系が暗黙的に

並列化を行うモデル（右）では，プログラミングが容

易になる反面，ハードウェアを活かしきった高性能

な処理が難しくなる．一方，データの送受信まで明

示的に記述するMPIのような並列化モデル（左）は，

高性能だがハードウェア構成を意識したプログラミ

ングが必要になってしまう．

　X10は，この 2つの中間の「PGAS（Partitioned 

Global Address Space）プログラミングモデル」を採

用している．このモデルでは，グローバルなアドレ

図 -1　X10 の実行モデル

☆ 6
 Unified Parallel C.

☆ 7
 Co-Array Fortran.

☆ 8
 Compute Unified Device Architecture.

おり，アクティビティがそのプレースに移動するこ

とでオブジェクトの中身にアクセス可能となる．

　X10ではさらに，複数のプレースにまたがった

特殊なデータ構造として「分散配列（DistArray）」

を作ることができる．分散配列の各要素は特定のプ

レースに所属しており，そのプレース内のアクティ

ビティによって操作される．

　X10はデータのアクセスに関して，「変更可能

（mutable）なデータは特定のプレースに所属してお

り，同じプレースのアクティビティからしかアクセ

スできない」という基本ポリシーで設計されている．

一方，生成後は変更できない変更不能（immutable）

データは，どのプレースからでもアクセス（読み出

し）することができる．これには，クラスや，後述

する構造体，関数などがある．

 ❏ Hello World
　図 -2は，X10で書いた Hello Worldプログラム

の例である．4行目の async文により，新しいア

クティビティが生成され，続く at文により指定さ

れたプレースへと移動する．そして，各プレース

で println文が実行され，プレースの ID（here.

id）が出力されている．

　図 -2 の下部は，このプログラムをコンパイル・

実行した例である．X10プログラムは現在，Java，

C++，CUDA
☆ 8などの環境に向けてコンパイル

可能で，それぞれ使用するコマンドが異なっている

ス空間が複数の「プレース（Places）」に分割されてい

る．プレースはメモリの局所性を抽象化したもので，

典型的には 1つのマシンに対応すると考えればよ

い．データはどれかのプレースに所属し，それをま

たいで移動することはない．データは他のプレース

からも参照することができるが，中身にアクセスで

きるのは同じプレースからのみである．

　X10は PGASを採用し，並列分散環境を隠蔽

するのではなく「抽象化して見せる」ことで，生産

性を保ちつつ高性能を達成することを目指してい

る．なお，PGAS自体は X10特有の概念ではなく，

UPC
☆ 6や CAF

☆ 7などの言語でも採用されてい

る．ただし，X10では以下に述べるように非同期

実行のサポートなどを強化しているため，区別して

APGAS（Asynchronous PGAS）モデルと呼ぶこと

もある．

 ❏ X10の実行モデル
　X10の実行モデルについて詳細に示したものが

図 -1下である．アドレス空間を分割した各プレー

スの中には，それぞれ複数の「アクティビティ」と

「オブジェクト」が存在できる．

　アクティビティは，プレース内で逐次動作する非

同期な実行主体で，軽量なスレッドだと考えればよ

い．後述する async文により，同じプレース内に

動的に生成可能である．また，at文により他のプ

レースに移動することができる．同じプレース内の

アクティビティ間では同期・排他制御を行うことが

できるが，他のプレースのアクティビティに対して

は，処理完了を待ち合わせる以外の同期は行えない．

　オブジェクトは特定のプレースに所属する変更可

能なデータ構造で，中身にアクセスできるのは同じ

プレースのアクティビティに限定される．ただし，

後述する GlobalRefという仕組みにより，他のプ

レースからオブジェクトを「リモート参照」すること

ができる．リモート参照には，参照しているオブジ

ェクトがどのプレースに存在するかの情報が入って

1 class MyHello {
2 public static def main(Array[String]) {
3 finish for (pl in Place.places()) {
4 async at (pl) Console.OUT.println(
5 "Hello World from place " + here.id);
6 } } }

$ x10c MyHello.x10 #
$ X10 NPLACES=4 x10 MyHello
Hello World from place 3
Hello World from place 0
Hello World from place 1
Hello World from place 2

#

図 -2　X10 による Hello World と，その実行例

☆ 9
 変数の型は，コンパイル時に推論（infer）される．型指定を省略しても，実行時の変数の型が動的になるわけではない．

名でなく thisを用いる（4行目）．

　変数やフィールドの宣言は，valもしくは var

というキーワードを用い，型名は変数名の後にコロ

ンで指定する（10行目など）．valは変更不能な変

数の宣言に用いるが，初期値から型が推論できる

場合は型の指定を省略できる☆ 9（9行目など）．フ

ィールドやメソッドには，Javaと同様，publicや

staticなどの修飾子が指定可能である．

　X10の特徴として，クラスやインタフェースの

ほかに，構造体や関数（クロージャ）を型として使用

可能なことが挙げられる．構造体は，値渡しされる

変更不能なデータのまとまりで，コンストラクタの

呼び出しは newなしで指定する（19行目）．関数は

「=>」を用いたリテラル表現（20行目）や，メソッド

セレクタ記法により生成され，「( )」により引数を

指定して実行される（21行目）．

　クラス，インタフェース，構造体は，総称型とし

て宣言することもできる（2行目）．型パラメータは

「[ ]」で指定し，実体化の際には必ず指定しなけれ

ばならない（15行目など）．

　制御構文や演算子には，Javaと同様のものが使

用できる（11行目など）．さらに，図 -2 で示した

asyncや atのような並列・分散処理のための構文

が，ここでは取り扱いが簡単な Java版を使ってコ

ンパイル・実行する例を示した．なお，プレース数

は構成によって変化するがこの実行例では 4つであ

る．アクティビティは並列に動作するため，出力の

順序は一定でないことに注意してほしい．

　次章以降では，X10のデータ型や文法について

解説していく．

X10 の基本文法

　X10は，静的に型付けされたオブジェクト指向言

語で，逐次処理部分の記法は Javaに近いものにな

っている．本章では，Javaとの違いを中心に X10

の基本文法を説明する．並列・分散処理については

次章で述べる．

 ❏ X10プログラミング入門
　図 -3は，X10の記法を概観するためのサン

プルプログラムである．X10プログラムの実行

は Javaと同様，指定したクラスの mainメソッ

ドから開始される（7行目）．ただし，引数の型は

Array[String](1)となる．ここで，「[String]」

は総称型クラス Arrayの型パラメータを示し，

「(1)」は配列の次元が 1という制約を表している．

なお，配列のアクセスは「( )」を用いて行われる（9

行目）．

　コメントは Javaと同様に「/*～*/」か「//～」の

形式で記述できる．また，パッケージ機構も Java

と同様のものが利用可能で，ライブラリを import

して使用することができる（1行目）．なお，x10.

lang.*と x10.array.*は自動的に importされ

るので指定しなくてよい．

　Javaと異なり，メソッドは defというキーワー

ドで宣言する（4～ 7行目）．返り値の型はコロンの

後に指定するが，推論できる場合は省略可能である．

また，メソッド本体を「=」の右辺として記述するこ

ともできる（5行目）．コンストラクタの宣言には型

図 -3　X10 の記法を概観するための例

1 import x10.util.Pair; // struct to represent a pair
2 public class MySample[T] {
3 val data:T;
4 def this(d:T) { data = d; } // constructor
5 def get() = data;
6
7 public static def main(args:Array[String](1)):void {
8 /* Add from 1 to arg */
9 val end = (args.size > 0) ? Int.parse(args(0)) : 10;
10 var sum:Int = 0;
11 for (var i:Int = 1; i <= end; i++) sum += i;
12 Console.OUT.printf("Sum of 1-%d: %d\n", end, sum);
13
14 /* Object creation with generics */
15 val obj = new MySample[Double](1.2);
16 Console.OUT.println(obj.get()); // -> 1.2
17
18 /* Various data types */
19 val pair = Pair[Int,Int](3,4); // struct
20 val func = (i:Int,j:Int)=>i*j; // function
21 Console.OUT.println(func(pair.first,pair.second)); // -> 12
22 }
23 }

☆ 10
 厳密には，true，false，nullは Javaのキーワードではなくリテラルだが，ここでは便宜上予約語としている．

　そのほかに，"Hello\n"は String型，[1,3, 

7] は Array[Int] 型，1..10 は Region 型 の

リテラルを表す記法である．また，「(i:Int, 

j:Int):Int=>i*j」のように書くことで「(Int, 

Int)=>Int」型の関数リテラルを表現することがで

きる．

制御構文と演算子

　X10 で は，if, else, switch, case, default, 

for, while, do, break, continue, throw, try, 

catch, finally, returnなどの制御構文を Java

と同様のイメージで使うことができる．ただし，

forによる Iterable要素のイテレーションは

「for (i in iterableData) ～」のように inを

使って記述する．ほかに X10特有の制御構文とし

て，並列・分散処理を記述するためのものがあるが，

それらについては次章で説明する．

　式や演算子についても Javaと同様のものが使え

るが，キャストを「v as TypeName」の形で記述す

る点が異なる．また，X10特有の演算子として，関

数リテラルや関数型の定義に使う「=>」，Region型

のリテラル生成に使う「..」，特定プレースへの分

散（Dist）を作る「->」，分散や分散配列の要素制限

を行う「|」，型の包含関係を調査または指定するの

に使う「<:」「:>」などがある．表 -1は，演算子の

が追加されている．

 ❏基本型と制御構文
　図 -4は X10の予約語一覧である．比較のため，

Javaの予約語も掲載している☆ 10．X10の予約語は

Javaとかなり重なっているが，並列・分散実行の

ためのキーワード等が追加されている．

基本型とリテラル

　X10では基本型は予約語ではない．Boolean, 

Char, Byte, Short, Int, Long, Float, Double, 

UByte, UShort, UInt, ULongが用意されているが，

これらは標準ライブラリ内で構造体として定義され

ている．符号なしの数値型が用意されている点も，

Javaと異なる点である．X10の基本型は，大文字

で始まる名前（たとえば Int）が正式であるが，標準

で読み込まれる x10.lang._クラス内で小文字で

始まる名前が type宣言されているため，そちら（た

とえば int）を使うこともできる．

　基本型にはそれぞれリテラル表現が存在している．

記法は Javaとほぼ同じであるが，Short型，Byte

型のリテラルも，それぞれ末尾に s，yを付ける

ことで表現可能である．また，符号なし型のリテ

ラルを表すためには末尾に uを付ける．たとえば

0xBBUYは UByte型のリテラル（値は 187）である．

図 -4　X10 と Java の予約語

☆ 11
 この挙動は X10 2.1からである，詳しくは「プレース」の節で述べる．

る☆ 11．

　オブジェクトの元となるクラスは，Javaと同じ

く class文によって定義され，extendsによる単

一継承により，x10.lang.Objectからの木構造を

成す．抽象（abstract）クラスや拡張不可（final）

なクラス，インタフェース（interface）も，Java

と同様に定義可能である．ただし，X10ではクラ

スの静的フィールドには，変更不能なもの（static 

valフィールド）しか定義できない．これは，どの

プレースからでもアクセス可能なデータは変更不能

というポリシーによる．

構造体

　構造体は，「値渡し」されるひとまとまりのデー

タを表すための型で，class文の代わりに struct

文で定義される（図 -6の 1行目）．この例では後述

するプロパティ機能を使って構造体 MyPairのフィ

ールド firstと secondを定義している．

　X10の構造体データは，内部を部分的に変更す

ることはできない．つまり，valフィールドしか

持てない．そのかわり，プレースをまたいで自由

に読み出し可能である．構造体は extendsで拡張

一覧と優先順位を示したものである．

　一部の演算子は，ユーザが定義することも可能で

ある．図 -5は，二次多項式を表す MyPolyという

クラスに「+」などの演算子をユーザ定義した例であ

る．defの代わりに operatorキーワードを用い，

定義したい演算子が出てくる部分にそれを書いて宣

言すればよい（5行目）．静的メソッドとして定義（6

行目）したり，暗黙の型変換を定義（11行目）するこ

ともできる．

 ❏データ型
　X10では，オブジェクト（class），構造体（struct），

関数（function）の 3種のデータ型を扱える．本節で

は，これらの概要と関係について，図 -6のサンプ

ルプログラムを参照しつつ説明する．

オブジェクト（クラス）

　オブジェクトは，Javaのオブジェクトとほぼ同

じで，クラス名を指定して newにより生成され，「参

照」を通じて使用される．オブジェクトは生成され

たプレースから移動できないが，アクティビティ

が他のプレースに移動する際に自動的に複製され

表 -1　演算子の優先度と結合規則

図 -5　演算子のユーザ定義例

1 public class MyPoly(a:Int,b:Int,c:Int) { // denotes a*x^2 + b*x + c
2 public def toString() = ""+a+"*x^2 + " + b+"*x + " + c;
3 operator this(n:Int) = a*n*n + b*n + c;
4
5 operator this+(p:MyPoly) = new MyPoly(a+p.a, b+p.b, c+p.c);
6 static operator (p:MyPoly)*(q:MyPoly) {
7 if (p.a!=0 || q.a!=0) Console.OUT.println("unsupported");
8 return new MyPoly(p.b*q.b, p.b*q.c+p.c*q.b, p.c*q.c);
9 }
10 operator -this = new MyPoly(-a,-b,-c);
11 static operator (n:Int):MyPoly = new MyPoly(0,0,n); // coercion
12
13 public static def main(Array[String]) {
14 val X = new MyPoly(0,1,0); // == 0*x^2 + 1*x + 0
15 val t = -X + 3; // -X+3
16 val u = 2*X + 1; // 2X+1
17 val v = t * u; // (-X+3)(2X+1)
18 Console.OUT.println(v); // -> -2*x^2 + 5*x + 3
19 Console.OUT.println(v(2)); // -> 5
20 val func = MyPoly.+; // operator function
21 Console.OUT.println(func(t,u)); // -> 0*x^2 + 1*x + 4
22 }
23 }

☆ 12
 X10 2.1.1からは，newを付けても呼べるようになった．

☆ 13
 ただし，Java版の X10では今のところ，基本型以外の構造体は Javaオブジェクトにマップされているので，メモリ消費上のメリットはない．

☆ 14
 関数リテラルの返り値の型は右辺の処理内容から推論されるが，明示したい場合は左辺の「)」の後にコロンをつけて指定することもできる．

 例：「(i:Int,j:Int):Int=>i*j」．

関数

　関数は，他の言語ではクロージャやラムダ式とも

呼ばれているデータである．X10の関数はオブジ

ェクトや構造体と同じく一級データなので，変数に

代入したり，引数として受け渡しも可能である（20

行目など）．構造体と同じく中身の変更はできない

が，プレースをまたいで使用可能である．関数の呼

び出しは，メソッドと同様に名前の後ろに「( )」で

引数を指定して行う（21行目など）．

　関数のリテラルや型は，「(a1, a2,…)=>b」の形式

で記述することができる（括弧は省略できない）．関

数リテラルでは，左辺の aの部分に仮引数の名前

と型，右辺の bの部分に処理内容を書く☆ 14．関

数の型を記述する場合は，左辺に引数の型名，右

辺に返り値の型名を書く．たとえば 11行目の

「(i:Int)=>i*x」は，引数として渡された Int値

を外部環境の x（関数定義中で束縛されていない変

数，10行目で宣言）と掛けて返す関数で，その型は

「(Int)=>Int」（Int値を 1つ受け取り Int値を

返す関数）である．右辺に「{ }」ブロックを用いて，

変数や制御構文を使った長い関数リテラルを記述す

ることもできる（23～ 27行目）．

　関数データはほかに，メソッド名の後ろに「.」と

引数の型を指定する「メソッドセレクタ記法」や型

名の後ろに「.」と二項演算子を指定する「オペレー

タ関数記法」によって作ることもできる．18行目の

「Math.pow.(Double,Double)」や，図 -5 の 20

行目の「MyPoly.+」はその例である．

　X10の関数型は無名クラスのような扱いで，型

を定義する functionのようなキーワードはない．

そのため，型パラメータを使用したり newで関数

データを生成することはできない．一方，関数型は

クラスや構造体で実装（implements）することがで

きる（2行目）．そのようなデータを関数として使用

する（9行目）と，引数の型に対応する「this(～)」

演算子（3行目）が呼び出される☆ 15．

することはできないが，メソッドや演算子を定義

する（3行目）ことや，インタフェースや関数を実装

（implements）する（2行目）ことは可能である．構

造体データの生成は，newなしでコンストラクタを

呼び出す☆ 12ことで行われる（8行目，11行目）．

　構造体は参照をとることができないので，オブジ

ェクトの nullに相当する値はない．オブジェクト

内に構造体のフィールドを定義した場合，構造体は

そのオブジェクト中にインラインされる．また，構

造体データの同一性（==）は内容が同じかどうかで

判断される．構造体は，長いビット数のスカラー値

だと考えると分かりやすいだろう．X10では，Int

や Doubleなどの基本型も，構造体として定義され

ている．

　構造体データにはメタ情報が不要なので，オブジ

ェクトよりもメモリ消費が少ないという利点があ

る☆ 13．一方，継承ができない，参照がとれず常に

値渡しである，などの制約があるので，それほど多

くない量のデータをまとめて扱いたい場合に用いる

のがよいだろう．

図 -6　構造体や関数の使用例

1 public struct MyPair[T,U](first:T, second:U)
2 implements (String)=>String {
3 public operator this(s:String) = s+"("+first+","+second+")";
4 static def curry[S](f:(S,S)=>S, b:S) = (x:S)=>f(x,b);
5
6 public static def main(Array[String]) {
7 /* Struct example */
8 val p = MyPair[Int,Double](1,2.3);
9 Console.OUT.println(p("Data is ")); // -> Data is (1,2.3)
10 val x<:Int = 4; // exact type is Int{self==4}
11 val q = MyPair[(Int)=>Int, Int]((i:Int)=>i*x, 5);
12 Console.OUT.println(q.first(q.second)); // -> 20
13 var a:Any = p;
14 Console.OUT.println(a.typeName());
15 // -> MyPair[x10.lang.Int,x10.lang.Double]
16
17 /* Function example */
18 val power = Math.pow.(Double,Double); // method selector
19 Console.OUT.println(power(2,5)); // -> 32.0
20 val square = curry[Double](power, 2);
21 Console.OUT.println(square(5)); // -> 25.0
22
23 val sq = (n:Int)=>{
24 var s:Int=0; val abs_n = n<0 ? -n : n;
25 for ([i] in 1..abs_n) s+=abs_n;
26 s // return value of the closure
27 };
28 Console.OUT.println(sq(5)); // -> 25
29 }
30 }

☆ 15
 実はこの演算子は implementsを宣言していなくても呼べる（「配列」の節を参照）が，宣言した場合は対応する this(～)演算子を実装していない
とコンパイルエラーになる．

☆ 16
 propertyキーワードは，「プロパティメソッド」の定義にも使用されるが，本稿では省略する．

クラスや構造体，インタフェースの定義では型パ

ラメータを指定することができる（1行目）．たとえ

ば 8 行目の MyPair[Int,Double] は，Int 値と

Double値のペアを表す型となる．型パラメータに

は，関数を含むあらゆる型を使用することができる

（11行目など）．また，メソッド定義に型パラメー

タを指定することも可能である（4行目）．

　Javaの総称型との主な違いとしては，型パラメ

ータの指定法が「< >」ではなく「[ ]」であること，

実装が型消去ではなく各データが型パラメータの実

際の値を保持していること（15行目），そのため使

用時に型パラメータを省略することはできないこと，

などが挙げられる．

プロパティと制約

　クラスや構造体，インタフェースの定義では，型

名の後ろに「( )」を付加して「プロパティ」を宣言

することができる．たとえば図 -6 の 1行目では，

MyPair構造体に対して firstと secondという

プロパティ（型はそれぞれ Tと U）を宣言している．

プロパティは変更不能なインスタンスフィールドで，

コンストラクタ内で property
☆ 16文でまとめて設

定されるが，ここで示した例のようにコンストラク

タ定義ごと省略することも可能で，その場合はプロ

型の階層構造

　図 -7は，クラス，構造体，関数，およびインタ

フェースの間の階層構造を示したものである．実

線の矢印は継承（extends），破線の矢印は実装

（implements）を表している．クラス部分について

は，Javaと同様である．クラスは単一継承による

階層構造をとり，すべて x10.lang.Objectのサ

ブクラスである．

　図中，太線で示されているのが，X10特有の部

分である．X10では Objectの上に x10.lang.

Anyという型（インタフェース）がある．構造体や関

数型は，すべて Anyの直下にあり，それらの間の

継承関係はない．ただし，クラスや構造体は複数の

インタフェースや関数を実装することができる．

　すべての型は，Anyインタフェースを暗黙的に実

装している．そのため，X10のあらゆるデータは

Any型の変数に代入可能である（13行目）．Anyには，

toString，typeName，equals，hashCode の 4

つのメソッドが宣言されており，どのデータに対し

てもこれらのメソッドを呼び出すことができる（14

行目）．

総称型

　X10では総称型（ジェネリクス）も利用可能で，

図 -7　型の階層構造

☆ 17
 静的に判定できない場合は，実行時にチェックするコードが（警告つきで）生成される．オプション指定でコンパイルエラーとすることもできる．

☆ 18
 どうしても型を明示したい場合は，図 -6 の 10行目のように「<:」を用いて「上限」を指定することもできる．

☆ 19
 この機能は，X10 2.1.1までは setおよび applyというメソッドによって実現されていた．

ティは，配列などの次元を表す「rank」であろう．

たとえば，「Array[String]{rank==1}」は，1

次元の文字列配列を示す型である．これは，図 -3

の 7行目のように「Array[String](1)」と書くこ

ともできる．これは，標準で読み込まれる x10.

lang._クラス内で，typeという構文を使って，

  public static type Array[T](r:Int)

              = Array[T]{self.rank==r};

と別名定義されているからである．

 ❏配列
　X10には，Javaのような言語コンストラクトと

しての配列は存在しない．しかし，「this(～)=～」

および「this(～)」という演算子により，配列アク

セス風の処理を一般化した形で実現している☆ 19．

前者が配列要素の設定，後者が配列要素の読み出し

に対応する処理を行う．図 -8は，これらの演算子

をユーザ定義することで連想配列を実装した例であ

る．3行目で演算子「this(String)=T」，4行目で

演算子「this(String)」を定義し，9～ 10行目で

配列アクセス風の呼び出しを行っている．

　X10 の標準クラスである x10.array.Array

は，同様の仕組みを使って配列を提供している．図

の 14行目からは，その使用例である．14～ 15行

目で，要素数 1000の 1次元整数配列を作成し，1

～ 1000の値を設定している．ここで内部的には

「this(～)=～」演算子が呼ばれる．16～ 17行目

では，この配列の各要素を二乗して合計する処理

を行っている．ここでは「this(～)」演算子が呼ば

れる．

　X10の Arrayは多次元配列もサポートしている．

また，添え字は 0から始まらなくてもよい．これ

らは，以下で述べる x10.array.Pointと x10.

array.Regionというクラスを用いて実現されて

いる．

パティ値を引数にとるコンストラクタが自動的に用

意される．

　X10では，型やメソッドに「{ }」を付加して

「制約」をつけることが可能で，プロパティはこ

の制約の記述に用いることができる．たとえば，

MyPair[Int,Int]{first!=0}は，第一要素の

値を 0以外に制限した MyPair型である．また，値

「4」の最も厳密な型は Int{self==4}である．こ

こで，selfは制約記述の中で用いる自分自身を表

すキーワードである．

　制約が満たされているかどうかは，基本的にコン

パイル時に型推論によりチェックされる☆ 17．X10

の型推論は非常に強力で，制約も含めた推論が行わ

れる．そのため，val変数の定義では型を明示的に

書かず初期値から推論させた方が，より厳密な型が

設定されて都合がよいことが多い☆ 18．

　X10プログラミングで最もよく使われるプロパ

1 public class MyAssoc[T] { // simple associative array
2 val map = new x10.util.HashMap[String,T]();
3 operator this(key:String)=(value:T) = map.put(key, value);
4 operator this(key:String) = map.getOrThrow(key);
5
6 public static def main(Array[String]) {
7 /* Array-like access */
8 val aa = new MyAssoc[Int]();
9 aa("one") = 1; // ==operator this("one")=1
10 aa("two") = 1 + aa("one"); // ==operator this("one")
11 Console.OUT.println(aa("two")); // -> 2
12
13 /* Basic array example */
14 val arr = new Array[Int](1000);
15 for ([i] in 1..arr.size) arr(i-1) = i;
16 var s:Int = 0;
17 for (pt in arr) s += arr(pt)*arr(pt);
18 Console.OUT.println(s); // -> 333833500
19
20 /* Point */
21 val pt1 = [1,3,7] as Point; // rank==3
22 Console.OUT.println(pt1(1)); // -> 3
23 val [a,b,c] = pt1; // a==1, b==3, c==7
24 val pt2[x,y] = Point.make(1,3); // x==1, y==3
25 Console.OUT.println(pt2); // -> [1,3]
26
27 /* Region */
28 val R1 = 3..5; // rank==1
29 val R2 = (-1..1)*R1; // rank==2
30 Console.OUT.println(pt2 in R2); // -> true
31
32 /* Array */
33 val arr1 = new Array[Int](R2, ([i,j]:Point)=>i*j);
34 Console.OUT.println(arr1(pt2)); // -> 3
35 Console.OUT.println(arr1(-1,4)); // -> -4
36 }
37 }

図 -8　配列の使用例

☆ 20
 この記法の「[ ]」を取ると，Intでなく Pointに対するイテレーションになる．

Arrayクラス

　Array[T]は要素の型が Tの配列を表すクラスで，

その定義域は Regionで指定される．Regionを用

いることで，X10では多次元配列や添え字が負の

配列，要素が疎な配列，三角行列なども柔軟に表現

できるようになっている．

　たとえば 33行目の arr1は，29行目で定義し

た 2次元 Region R2を定義域とする整数配列であ

る．Arrayにはさまざまなコンストラクタがあるが，

ここでは第 2引数に (Point)=>T型の関数をとる

ものを用い，生成時に初期値を計算・設定するよう

にしている．

　配列の各要素には，定義域に含まれる Point

や，各次元の Int値を直接指定することでアクセ

スできる（34～ 35行目）．Region外のインデック

スを使用すると，x10.lang.ArrayIndexOutOf  

BoundsExceptionとなる．

　なお，Arrayのサイズ（要素数）は Javaと同

じく変更不能で，sizeプロパティに入っている

（15行目）．また，Arrayや Regionは Iterable 

[Point]を実装しているので，含まれる Pointを

for文で順に取り出すことができる（17行目）．

並列・分散プログラミング

　前章では，X10の基本的な文法とデータ型につ

いて，Javaと比較しながら解説したが，使用した

例はいずれも，1プレース・1アクティビティで逐

次実行されるものであった．しかし，X10の本領

は，複数アクティビティを用いた並列処理や複数プ

レースにまたがった分散処理を容易に記述できる点

にある．

　本章では，並列・分散処理のための機能と，それ

らを利用するための特徴的な構文やデータ構造につ

いて解説する．

Pointクラス

　Pointは n次元整数格子上の 1点を表すクラス

で，n次元配列の「添え字」として使用できる．たと

えば，図中 21行目の pt1は 3次元上の点 (1, 3, 7)

を表している．Point型のデータは，整数配列か

らキャストしたり（21行目），Point.makeメソッ

ド（24行目）で作成できる．各次元の要素は Int型

に限定されており，負の値も可能である．pt1(1)

のようにして読み出せる（22行目）が，個別に変更

することはできない．なお，次元数 nは rankプロ

パティとして保持されており，制約の指定にも使用

できる．

　Pointでは，「分解型」のアクセス法（destructuring 

syntax）がサポートされている．これは，変数宣言

時に「[ ]」を使って指定される．23～ 24行目がそ

の例で，変数 a，b，c，x，yに Pointの各次元の

値が設定される．実は，「for ([i] in 1..10) 

～」のような記法（15行目など）はこれを利用したも

のである☆ 20．

Regionクラス

　Regionは次元（rank）が同じ Pointの集合

を表すクラスで，配列の「有効な添え字（定義

域）」を示すのに使用できる．28行目の「3..5」

は，1次元の Point (3) (4) (5)の 3点からなる，

Region{rank==1}型のリテラルである．Region

クラスではさまざまな演算が定義されているが，よ

く使うのは「*」による直積である．29行目の R2は，

(21, 3) (21, 4) (21, 5) (0, 3) (0, 4) (0, 5) (1, 3) (1, 

4) (1, 5)の 9点からなるRegion{rank==2}となる．

　なお，Regionには必ずしも実際の Pointが入

っているわけではなく，集合の「形」が論理的に記録

されている．ある Pointが Regionに含まれるか

どうかは，in演算子や containsメソッドで調べ

ることができる（30行目）．

が終了するとこの finish文を抜け，11行目で結

果が合計される．

　finish文ではその内部のすべてのアクティビテ

ィの終了を確定できるので，これを try～catch

で囲むことで，非同期処理中に起きた例外もキャ

ッチすることができる．これを「Rooted Ex cep tion 

Model」と呼ぶ．複数のアクティビティで例外が起

きた場合は，x10.lang.MultipleExceptions

が送られる．

　X10の async文は，ハンドルのようなものは返

さない．そのため，アクティビティの終了を待ち合

わせることができるのは，親アクティビティかその

先祖だけであり，待ち合わせがループすることはな

い．これにより，意図せぬデッドロックを言語仕様

レベルで抑制している．アクティビティは，main

を実行開始する「初期アクティビティ」をおおもとと

する木構造を成し，それらが全部終了すると X10

プログラムが終了する．mainの呼び出しの外側に

仮想的な finish文があると考えればよい．

　asyncブロック内では，その外側で定義された

ローカル変数を読み書きすることができる（28行

目）．ただし，複数のアクティビティから同時に書

き込まないように注意が必要である．そのための排

他制御については，節をあらためて述べる．

 ❏プレース
　プレースはメモリを共有する 1つの実行環境を

抽象化したもので，典型的には個別のマシンに対応

すると考えればよい．アクティビティは，at文で，

別のプレースに移動することができる．22行目は，

隣のプレース（here.next()）でフィボナッチ数を

計算する例である．ここで，hereは「現在のプレ

ース」を示す x10.lang.Place型構造体の変数で

ある．プレースには 0から Place.MAX_PLACES21

の続き番号が振られており，idプロパティとして

アクセスできる（図 -2 の 5行目）．

　at文では，新しいアクティビティが生成される

 ❏アクティビティ
　「実行モデル」の節で述べたように，X10の実行

環境は複数の「プレース」から構成され，プレース内

では複数の「アクティビティ」を実行できる．アクテ

ィビティの生成は，async文で行われる．図 -9の

1～ 18行目は，複数のアクティビティを生成して

フィボナッチ数を並列計算するプログラムの例で

ある．8行目の async文で 1つ前のフィボナッチ

数を並列計算している間に，9行目で 2つ前のフィ

ボナッチ数を計算している．7行目の finish文は，

そのブロック自身と内部で生成したアクティビティ

（孫も含む）がすべて終了するまで待つ文である．ア

クティビティは再帰的に多数生成されるが，すべて

1 public class MyFib {
2 var r:Int; // in-out parameter
3 def this(i:Int) { r = i; }
4 def run() {
5 if (r < 2) return r; // MyFib(0)==0, MyFib(1)==1
6 val f1 = new MyFib(r-1), f2 = new MyFib(r-2);
7 finish {
8 async f1.run(); // compute MyFib(r-1) in parallel
9 f2.run(); // compute MyFib(r-2) by myself

10 }
11 return (r = f1.r + f2.r); // MyFib(r-1) + MyFib(r-2)
12 }
13
14 public static def main(args:Array[String](1)) {
15 /* Compute a Fibonacci number in parallel */
16 val n = (args.size>0)? Int.parse(args(0)) : 10;
17 val f = new MyFib(n); f.run();
18 Console.OUT.println("Fib("+n+")="+f.r); // -> Fib(10)=55
19
20 /* Place shift */
21 Console.OUT.println(Place.MAX_PLACES); // -> 4 (etc.)
22 val r = at (here.next()) new MyFib(10).run();
23 Console.OUT.println(r); // -> 55
24
25 /* Local variable access */
26 var m:Int = 0; // mutable
27 val i = 1; // immutable
28 finish async { m = i; } // ok
29 Console.OUT.println(m); // -> 1
30 val h = here;
31 at (here.next()) {
32 Console.OUT.println(i); // -> 1
33 // m = 2; // compile error
34 at (h) m = 2; // ok
35 }
36 Console.OUT.println(m); // -> 2
37
38 /* Object and GlobalRef */
39 val o = new MyFib(0);
40 at (here.next()) o.r = 1; // copy of o is modified
41 Console.OUT.println(o.r); // -> 0
42 val g = GlobalRef[MyFib](o);
43 at (here.next()) { at (g.home) g().r = 2; }
44 Console.OUT.println(o.r); // -> 2
45 }
46 }

図 -9　アクティビティとプレースの使用例

☆ 21
 この挙動は X10 2.1からのもので，将来的に Javaオブジェクトを X10オブジェクトと区別せず扱えるようにするための準備として導入された．
オブジェクトのフィールドを transientと宣言することで，値の複製を抑制することもできる．

16行目では，1000個のアクティビティを生成し

12
122

1...110002を計算しているが，sに値を足し

込む部分を atomic文により排他制御している（15

行目）ので，正しい結果が得られる．メソッド定義

に atomicをつけて，メソッド全体を atomicブロ

ックとして扱うこともできる（9行目）．不可分処理

のためにはほかに，x10.util.concurrentパッ

ケージのクラス（AtomicIntegerなど）を利用する

こともできる．

　「when (condition) ～」は，指定された条件

が成立するまでブロックする構文である．条件のチ

ェックと本体の実行は，不可分（atomic）に行われ

る．図 -10の MyBufは 1つの Int値を保持できる

わけではなく，意味的にはそのアクティビティ自体

が指定プレースに「移動」して処理が実行される．内

部の処理が終了すると元のプレースに処理が戻って

くるが，この際に値を返すこともできる（22行目）．

at内で生成したアクティビティの処理は終了して

いなくてもよいので，「at (place) async ～」

（もしくは「async at (place) ～」）で，指定プレ

ースでアクティビティを分散動作させることが可能

である．最初に示した図 -2 はその例である．

　概要でも述べたように，「変更可能なデータは特

定のプレースに所属しており，同じプレースからし

かアクセスできない」というのが X10の基本ポリシ

ーである．そのため atブロック内では，その外側

で定義された val変数を読むことはできる（32行

目）が，var変数にアクセスするには，それが宣言

されたプレースに戻る必要がある（33～ 34行目）．

　構造体や関数，クラスは変更不能なデータであり，

どのプレースからも自由にアクセスできる．一方，

オブジェクトは特定のプレースに所属しており，生

成されたプレースからしかアクセスできない．at

ブロック内からオブジェクトにアクセスすると，暗

黙のうちに複製が作られる☆ 21ため，取り扱いには

注意が必要である（40～ 41行目）．

　オブジェクトを x10.lang.GlobalRef[T]とい

う構造体に入れることで，暗黙の複製を抑制し，同

一オブジェクトへのグローバルな参照を持つことが

できる．ただし，中身にアクセスするには，オブジ

ェクトの存在するプレース（GlobalRefの homeプ

ロパティに入っている）に移動する必要がある．42

～ 43行目がその使用例である．

 ❏排他・同期制御
　同一プレースで動作するアクティビティの間で

は，さまざまな同期制御を行える．図 -10はその例

である．

　atomic文により，プレース内で不可分に実行

されるブロックを記述することができる．13～

図 -10　排他・同期制御の例

1 public class MyBuf {
2 var datum:Int = 0, filled:Boolean = false;
3 def send(v:Int) {
4 when (!filled) { filled = true; datum = v; }
5 }
6 def receive():Int {
7 when ( filled) { filled = false; return datum; }
8 }
9 static atomic def say(s:String) = Console.OUT.println(s);

10
11 public static def main(Array[String]) {
12 /* Atomic block */
13 var s:Int = 0;
14 finish for ([i] in 1..1000) async {
15 val tmp = i*i; atomic { s += tmp; }
16 }
17 Console.OUT.println(s); // -> 333833500
18
19 /* Conditional atomic blocks */
20 val buf = new MyBuf();
21 finish {
22 for ([i] in 0..9) async buf.send(i); // senders
23 for ([i] in 0..9) Console.OUT.print(buf.receive());
24 Console.OUT.println(); // -> 0217635489 (etc.)
25 }
26
27 /* Clocked execution */
28 clocked finish {
29 clocked async { // clocked activity A
30 say("A-1 "); next;
31 say(" A-2"); next;
32 say("A-3 ");
33 }
34 clocked async { // clocked activity B
35 System.sleep(1000); // sleep 1 sec
36 say("B-1 "); next;
37 say(" B-2"); next;
38 say("B-3 ");
39 }
40 clocked async { // clocked activity C
41 for ([i] in 1..3) {
42 val spc = (i%2==0) ? " " : "";
43 say(spc + "C-" + i); next;
44 } } }
45 } }

A-1
C-1
B-1

C-2
B-2
A-2

B-3
C-3
A-3

ラムは，デッドロックしない（できない）．

　1.	at，async，finish，atomicは自由に使っ

てよいが，whenは使わない．

　2. クロックは使ってよいが，finishブロックの

外側で生成したクロックを内側で使わない．

2番目の条件は分かりにくいが，図 -10 で示したよ

うに clocked finishと clocked asyncをセッ

トで使用し，途中に余計な finishをはさまなけれ

ば問題は生じない．

 ❏分散配列
　複数のプレースにまたがった分散処理のためのデ

ータ構造として，X10では，要素が各プレースに

散らばった「分散配列（x10.array.DistArray）」

を作成可能である．分散配列の各要素がどのプレー

スに存在するかは，分散（x10.array.Dist）デー

タによって示される．図 -11は，これらのクラスの

使用例である．

Distクラス

　Distは，Region内の各 Pointから Placeへ

のマッピングを表すクラスで，分散配列の各要素が

どのプレースに存在するかを示すのに使われる．

　最も一般的な Distの作成方法は，Dist.make 

Block(region)である．これにより，指定され

た Region内の点をプレース間で均等に分割した

Distが生成される．たとえばプレース数が 4の場

合，Dist.makeBlock(0..9) は， 点 (0) (1) (2)

をプレース 0，(3) (4) (5)をプレース 1，(6) (7)

をプレース 2，(8) (9)をプレース 3にマップする

Distになる（4行目）．

　特定の Pointがどのプレースにマップされて

いるかは，Dist変数を関数として呼び出すこと

で調べられる（6行目）．これは，Distクラスに

this(Point):Place や this(Int):Place とい

う演算子が定義されているからである．

　「|」演算子で，既存の Distから要素を制限した

バッファで，when文でバッファの状態 filledを

チェックして送り手と受け手の同期制御を行ってい

る（2～ 8行目）．22～ 23行目が使用例で，10個の

アクティビティから値を送り，元のアクティビティ

でそれらを受け取っている．

　アクティビティ間の同期のためにはほかに，「ク

ロック」という仕組みが用意されている．これは，

全タスクが特定の処理を終えるまで先に進むことを

抑止する「バリア同期」の概念を拡張して，複数のフ

ェーズに対して繰り返し使えるようにしたもので

ある．使用するにはまず，クロックに複数のアクテ

ィビティを「登録」する．各アクティビティは，1フ

ェーズ分の処理を終えると nextを実行する．する

と，登録されている全アクティビティが nextを呼

び出すまで，処理がブロックされる．

　28～ 44行目が，クロックの使用例である．28

行目の「clocked finish」により，クロックが生

成される．その内部の「clocked async」は，アク

ティビティを生成し，そのクロックに登録する構

文である．ここでは，3つのアクティビティ A，B，

Cを生成・登録している．各アクティビティでは，

順に 1，2，3を出力しているが，それぞれの間で

nextを呼び，バリア同期を行っている．そのため，

図の右下に示した実行例のように，A，B，Cの 3

つ分の出力が揃わないと次の出力へは進まない．

　クロックへの登録は，アクティビティが終了する

と解除される．また，通常の finish文と同様，内

部のアクティビティがすべて終了すると，clocked 

finish自体が完了し，クロックも解放される．

　複雑なバリア同期のためには，resume文でフ

ェーズ終了を「早めに伝える」ことや，x10.lang.

Clockオブジェクトを明示的に生成したり複数用

いることも可能であるが，本稿では省略する．

　並列プログラミングの難しさの 1つは，デッドロ

ックの回避である．X10では，デッドロックが起

きにくいように言語仕様レベルで工夫がなされてい

る．具体的には，以下の条件を満たす X10プログ

切な関数を渡すことで，1行で記述することができ

る（36行目）．

マルチコア時代と X10

　以上，本稿では X10の設計思想，基本的なデー

タ構造と文法，および並列・分散処理の概要につい

て解説した．より詳しい内容としては，http://x10-

lang.org/
1）から言語仕様 2）やチュートリアル資料を

入手できる．ソースコードは同サイトのリンクか

ら SVNで取得可能で，サンプルプログラムも多数

含まれている．日本語の資料としては，この解説

のほかセミナー資料 3）などがある．本稿ではクラス

ライブラリについてはあまり触れられなかったが，

HashMapなど基本的なものについては，Javaと同

新しい Distを作成することができる（7～ 8行目）．

また，「->」演算子で，Region内のすべての点を特

定のプレースへとマッピングすることができる（9

行目）．ほかに，Dist.makeUnique()では，存在

するプレース分の単純なマッピングが生成される

（10行目）．

DistArrayクラス

　DistArray[T]は要素の型が Tの分散配列を表

すクラスである．定義域と要素の存在場所は作成時

に Distで指定される（後から変更することはでき

ない）．Distを用いることで，X10では分散配列

の各要素とプレース（≒実マシン）との対応を柔軟に

制御することが可能になっている．

　DistArrayは，newではなく DistArray.make 

[T]というファクトリーメソッドで，Distを指定

して生成する．第 2引数に関数を渡して，生成時

に初期値を設定することもできる（15行目）．分散

配列の各要素にアクセスするには，その要素が存在

するプレースに移動する必要がある（17行目）．異

なるプレースからアクセスすると，x10.lang.

BadPlaceExceptionとなる．

　20～ 33行目は，DistArrayを用いて 12
122

1... 

110002の計算を分散して行うプログラム例である．

まず，1～ 1000の値が入った DistArray da0を

生成する（20～ 21行目）．そして，各プレースでの

計算結果を受け取る配列 tmpを用意し（22～ 23行

目），分散して計算を行う（24行目）．各プレースで

はまず，「|」演算子で da0の自プレース分の部分配

列を作り（26行目），つづく for文で各要素を二乗

して合計している（28行目）．全プレースの計算終

了を finish（24行目）で待った後，各プレースの

計算結果を合計して結果を得る（32行目）．

　実は，DistArrayには同様の処理を簡単に行う

ためのメソッドが用意されている．各要素に対し特

定の処理を行う mapメソッドと要素を特定の処理

方式で集約する reduceメソッドを組み合わせ，適

1 public class MyDist {
2 public static def main(Array[String]) {
3 /* Dist */
4 val D1 = Dist.makeBlock(0..9);
5 // (0..2)->0 (3..5)->1 (6..7)->2 (8..9)->3
6 Console.OUT.println(D1(3)); // -> (Place 1)
7 val D2 = D1 | here; // (0)->0 (1)->0 (2)->0
8 val D3 = D1 | 2..4; // (2)->0 (3)->1 (4)->1
9 val D4 = 0..9->here; // (0..9)->0
10 val D5 = Dist.makeUnique(); // (0)->0 (1)->1 (2)->2 (3)->3
11
12 /* DistArray */
13 val R = (1..4)*(5..6);
14 val D = Dist.makeBlock(R); Console.OUT.println(D);
15 val da = DistArray.make[Int](D, ([i,j]:Point)=>i*j);
16 for (pt in da)
17 Console.OUT.println(at (da.dist(pt)) da(pt));
18
19 /* Advanced DistArray example */
20 val D0 = Dist.makeBlock(1..1000);
21 val da0 = DistArray.make[Int](D0, ([i]:Point)=>i);
22 val places = da0.dist.places(); // Sequence[Place]
23 val tmp = new Array[Int](places.size);
24 finish for ([i] in 0..(places.size-1)) async {
25 tmp(i) = at (places(i)) {
26 val a = da0 | here; // restriction
27 var s:Int = 0;
28 for (pt in a) s += a(pt)*a(pt);
29 s // return value of at
30 };
31 }
32 var s0:Int = 0; for (pt in tmp) s0 += tmp(pt);
33 Console.OUT.println(s0); // -> 333833500
34
35 /* DistArray.map and reduce */
36 val s1 = da0.map((i:Int)=>i*i).reduce(Int.+, 0);
37 Console.OUT.println(s1); // -> 333833500
38 }
39 }

図 -11　分散配列の使用例

のプログラムで扱えるようになっている．プレース

間の通信は，at文や DistArrayによって抽象化

されており，プレースごとにアーキテクチャが異な

っていてもかまわない．そして，開発をサポートす

るツールとして，X10DTが提供されている．

　マルチコアの一般化により，専門家だけが並列・

分散プログラミングを行う時代ではなくなりつつあ

る．そのため，新しいプログラミングモデルと，言

語の基本コンストラクトとして並列・分散実行を指

定できる機能は必須となっていくだろう．それらを

備えた X10のような新言語が，マルチコア時代の

ソフトウェアの生産性と性能向上に役立つことを期

待してやまない．

様に利用可能である．これも上記サイトから仕様を

確認できる．

　X10は，C++か Java環境が用意されていれば

動作可能で，MPI，GPGPUや Blue Geneなども含

めたさまざまな環境で使用できるが，まずは試し

てみたいという場合は，開発ツールである X10DT

（X10 Development Toolkit）を使うのがよいだろう．

これは Eclipse用のプラグインで，個別インストー

ルすることもできるが，Eclipse実行環境も含んだ

ものが配布されているので，それを利用するのが簡

単である．図 -12は，X10DTの使用イメージである．

X10DT上で，X10プログラムの開発と実行を行う

ことができる．本稿に掲載したプログラムは，それ

ぞれ単体でコンパイル・実行が可能なので試しても

らいたい．

　さて，本稿の冒頭で，マルチコア時代のプログラ

ミング環境に求められる要件を挙げた．それらに

ついての X10の対応は以下のとおりである．まず，

非同期タスクは，async文により容易に生成可能で，

デッドロックが起きないような構文上の工夫が行わ

れている．また，実行モデルとして「APGAS」を採

用し，言語レベルで「プレース」という概念を導入す

ることで，共有メモリ環境と分散メモリ環境を単一

図 -12　X10DT の使用イメージ

